MATRIX EIGENVALUES

PURPOSE

Compute the eigenvalues of a matrix.

DESCRIPTION

A vector v is a eigenvector (or characteristic vector) of matrix A belonging to eigenvalue (or characteristic value) λ if

$$Av = \lambda v \quad \text{(EQ 4-65)}$$

or

$$(\lambda I - A)v = 0 \quad \text{(EQ 4-66)}$$

The eigenvalues are the solutions to the characteristic equation (det(λI-A)). The corresponding eigenvectors are computed by substituting the value of λ in the formula λI-A and solving for v.

SYNTAX

LET <var> = MATRIX EIGENVALUES <mat> <SUBSET/EXCEPT/FOR qualification>

where <mat> is a matrix for which the eigenvalues are to be computed;

<var> is a variable where the resulting eigenvalues are saved;

and where the <SUBSET/EXCEPT/FOR qualification> is optional and rarely used in this context.

EXAMPLES

LET C = MATRIX EIGENVALUES A

NOTE 1

If the matrix is symmetric, the eigenvalues are all real. If the matrix is not symmetric, the eigenvalues may be either real or complex. DATAPLOT handles this by putting the real components in the first N (where N is the number of rows in the matrix) rows of the returned variable and the corresponding complex components in rows N+1 through 2*N. If you prefer to have the real and complex components in separate variables, do something like the following:

```
LET E = MATRIX EIGENVALUES M
LET N2 = SIZE E
LET N = N2/2
LET N1 = N + 1
LET ER = E
LET EC = E
RETAIN ER FOR I = 1 1 N
RETAIN EC FOR I = N1 1 M
```

NOTE 2

DATAPLOT uses EISPACK to compute the eigenvalues. DATAPLOT uses the routines for either a real symmetric matrix or a real non-symmetric matrix. It does not handle complex matrices. Previous versions (before August 1993) used the routines from the Numerical Recipes book, which were limited to symmetric matrices.

NOTE 3

The matrix must have the same number of rows and columns. An error message is printed if this is not the case.

DEFAULT

None

SYNONYMS

EIGENVALUES for MATRIX EIGENVALUES

RELATED COMMANDS

MATRIX EIGENVECTORS	=	Compute the matrix eigenvectors.
CORRELATION MATRIX	=	Compute the correlation matrix of a matrix.
VARIANCE-COVA MATRIX	=	Compute the variance-covariance matrix of a matrix.
PRINCIPAL COMPONENTS	=	Compute the principal components of a matrix.

SINGULAR VALUES = Compute the singular values of a matrix.

REFERENCE

Any standard text on linear algebra.

APPLICATIONS

Linear Algebra

IMPLEMENTATION DATE

87/10 (conversion to LINPACK routines and extended to non-symmetric matrices 93/8)

PROGRAM 1

```
DIMENSION 100 COLUMNS
READ MATRIX X
16 16 19 21 20
14 17 15 22 18
24 23 21 24 20
18 17 16 15 20
18 11 9 18 7
END OF DATA
LET A = MATRIX EIGENVALUES X
PRINT A
```

The following output is generated.

```
VARIABLES--A

  0.8748484E+02
 -0.9631159E+01
  0.1899911E+01
  0.1899911E+01
 -0.5653489E+01
  0.0000000E+00
  0.0000000E+00
  0.1940936E+01
 -0.1940936E+01
  0.0000000E+00
```

PROGRAM 2

```
. PERFORM A PRINCIPAL COORDINATE ANALYSIS, A METRIC SCALING
. TECHNIQUE BASED ON EIGENVALUES AND EIGENVECTORS.
. SOURCE: "GRAPHICAL EXPLORATORY DATA ANALYSIS", DU TOIT, ET. AL. (pp 127-131).
. DATA MATRIX IS A DISSIMILARITY MATRIX FOR OCCUPATIONS. TECHNIQUE IS:
. 1) CALCULATE: Aij = (-1/2)*Dij**2
. 2) DOUBLE CENTER A BY SUBTRACTING ROW MEAN AND COLUMN MEAN, THEN ADD
.    GRAND MEAN
. 3)  FACTOR MATRIX INTO X*X-TRANSPOSE
. 4) X* = (SQRT(L1)*V1, SQRT(L2)*V2, ..., SQRT(Lk)Vk) WHERE THE L AND V ARE
.    THE EIGENVALUES AND EIGENVECTORS IS A K-DIMENSIONAL SCALING OF THE N
.    CASES.  IDEA IS TO PROVIDE A SCALING IN A LOW DIMENSION (1, 2, OR 3).  THE
.    RATIO OF THE EIGENVALUES USED (I.E., UP TO K) TO SUM OF ALL EIGENVALUES
.    PROVIDES A MEASURE OF THE GOODNESS OF FIT.
FEEDBACK OFF; DIMENSION 100 COLUMNS
READ MATRIX X
0.00 3.06 2.14 3.21 3.51 4.40 3.64 3.12 3.73
3.06 0.00 3.95 3.95 4.17 3.77 3.69 4.13 4.05
2.14 3.95 0.00 3.03 2.82 3.86 3.47 2.72 3.31
3.21 3.95 3.03 0.00 3.33 4.14 3.90 2.58 3.46
3.51 4.17 2.82 3.33 0.00 3.60 3.56 3.68 3.29
4.40 3.77 3.86 4.14 3.60 0.00 2.53 4.17 3.92
3.64 3.69 3.47 3.90 3.56 2.53 0.00 4.10 3.59
3.12 4.13 2.72 2.58 3.68 4.17 4.10 0.00 3.72
3.73 4.05 3.31 3.46 3.29 3.92 3.59 3.72 0.00
END OF DATA
LET P = MATRIX NUMBER OF ROWS X
. CALCULATE Aij = (-1/2)Dij**2 AND CALCULATE GRAND MEAN, COLUMN MEANS
. NOTE THAT SINCE X IS SYMMETRIC, ROW MEANS EQUAL TO COLUMN MEANS.
. DOUBLE CENTER MATRIX (Bij = Aij - Ai. - A.j + A.., where Ai.,
. A.j, and A.. ARE THE ROW MEANS, COLUMN MEANS, AND GRAND MEAN)
LOOP FOR K= 1 1 P
    LET X^K = X^K*X^K
    LET X^K = (-1/2)*X^K
    LET TEMP = MEAN X^K
    LET COLMEAN(K) = TEMP
    LET X^K = X^K - TEMP
END OF LOOP
LOOP FOR K = 1 1 P
    LET VTEMP = MATRIX ROW X K
    LET TEMP = COLMEAN(K)
    LET VTEMP = VTEMP - TEMP
    LET X = MATRIX REPLACE ROW X VTEMP K
END OF LOOP
LET GRAND = MEAN COLMEAN
LET X = MATRIX ADDITION X GRAND
. CALCULATE EIGENVALUES, EIGENVECTORS
LET E = MATRIX EIGENVALUES X
LET ESUM = SUM E
LET EVECT = MATRIX EIGENVECTORS X
. CALCULATE SQRT(L1)*V1, SQRT(L2)*V2, PLOT
LET L1 = E(P)
LET L1 = SQRT(L1)
LET V1 = L1*EVECT^P
LET TEMP = P - 1
LET L2 = E(TEMP)
LET L2 = SQRT(L2)
```

```
LET V2 = L2*EVECT^TEMP
LET RATIO = (L1*L1 + L2*L2)/ESUM
LEGEND 1 RATIO = ^RATIO
TITLE PRINCIPAL COORDINATE ANALYSIS
X1LABEL SQRT(L1)*V1
Y1LABEL SQRT(L2)*V2
LET TAG = SEQUENCE 1 1 P
CHARACTER 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25
LINE BLANK ALL
PLOT V2 V1 TAG
```